
Part 1: Creating Components 
ColdFusion components (CFCs) are special objects that let you encapsulate and reuse code in 

small, logical units. They contain data (also called properties) and functions (methods). Using 

components with ColdFusion allows developers to use a more object-oriented approach to 

their applications, allowing for encapsulation, inheritance, and introspection. As we learned 

earlier, CFCs allow you to group ColdFusion statements into user-defined functions, group 

related functions in a single CFC, and make your applications more robust and maintainable 

by separating complex business logic into a series of highly abstracted and structured code 

libraries.  

 

Benefits of CFCs include,  

 Developers can encapsulate their code into easily understood building blocks that are 

easily reused within and between applications 

 Allows for object-based development and use of object-oriented features like 

inheritance and data hiding 

 Keeps related functions within a single unit that can be secured using roles-based 

security 

 Can be called directly by Flex applications using AMF and as web services by many 

other applications and clients, including Flash, .NET, and PHP 

 Are self documenting (introspection) 

 Can persist between page requests 

 

A component contains two main elements: methods, which are the functions within the 

component, and properties, which is the data associated with it.  There are two basic types of 

components: 

 Static components - the simplest form which has one or more methods, no data 

structures bound to them, and are used minimally within advanced applications 

 Instance-based components - contain methods and properties, may utilize the 

properties that are associated with components by having persistent instances, and has 

data structures bound to them that are available to all methods as properties 

The basic structure of a CFC is: 

<cfcomponent> 

<cffunction name="function_name1" returntype="returntype"> 

<cfargument argument_name" /> 

[CFML code] 

<cfreturn expression_to_return> 

</cffunction> 

<cffunction name="function_name2" returntype="returntype"> 

<cfargument name="argument_name" /> 

[CFML code] 

<cfreturn expression_to_return> 

</cffunction> 

</cfcomponent> 

Component properties can be set using the THIS scope, and should be defined immediately 

after the opening <cfcomponent> tag 

<cfcomponent> 

<cfset this.datasource = "CF8advStepByStep"> 

... 

</cfcomponent> 

Functions 

As we noted before, functions, also called methods, are a series of ColdFusion statements that 

are grouped within a logical unit. Within a CFC, you can declare multiple functions. A function 

is declared using the <cffunction> tag. This tag gives you several benefits and features: 

 Validating the data type of returned data - functions generally return a single value the 

type of which must be predeclared using the returntype attribute 

 Access scoping that allows you to designate where a routine can be invoked from 

 Roles-based security that allows you restrict usage based on the roles defined within 

the <cflogin>/<cfloginuser> framework. 

 Controls whitespace by suppressing all the output generated and reducing the amount 

of extraneous spaces generated 

 Protected variable scoping allowing for local variables that will not overwrite like-named 

variables on the calling page and visa versa; data is instead passed through defined 

arguments or as part of the components instantiation using the init() method 

 Supports for easily returning data to AJAX applications, in JavaScript Object Notation 

(JSON) format, Web Distributed Data Exchange (WDDX) -- an XML format, or as plain 

text  

At minimum, the <cffunction> tag requires two attributes: 

 Name - the name of the function which is used to call it 

o It must be unique within the component and must follow ColdFusion's normal 

variable naming rules 

 Returntype - the datatype of the data that will be returned; allowed values include: 

o Any - don't use unless absolutely required :-P  

o Array 

o Boolean 

o Date 

o GUID 

o Numeric 

o Query 

o String 

o Struct 

o VariableName - a string formatted according to ColdFusion variable naming 

conventions 

o A Component Name - ColdFusion methods can return component instances. 

o Void (returns no value) 

Variables within a function should be set using the VAR attribute of the <cfset> tag, to avoid 

overwriting the contents of like-named variables in a different functions or global CFC 

variables. In this example, a local variable named datestr is declared and subsequently 

returned when the function currentdate is called. 

<cffunction name="currentdate" returntype="String"> 

<cfset var datestr = dateformat(now(), "mmmm dd, yyyy")> 

<cfreturn datestr> 

</cffunction> 

When using local variables in a function, they must be declared immediately after the 

<cffunction> and <cfargument> tags.  

 

As with CFML custom tags, functions cannot read the local variables on the calling page, nor 

can a function modify those variables. In order for it to process external data, functions must 

have values passed to them explicitly. Each parameter that may be passed to a function 

should be explicitly declared and validated within the function using the <cfargument> tag. 

They are access using the ARGUMENTS scope. 

<cffunction name="timesTen" returntype="numeric"> 

<cfargument type="numeric" name="x" required="true" /> 

<cfreturn arguments.x * 10> 

</cffunction> 

You may have noticed by now that while most ColdFusion variables are "loosely" typed, CFC 

use "strongly" type variables both in passing data to a function and in returning them. With 

strongly typed variables, ColdFusion will throw an error anytime the data passed does not 

match its declared type. For example, if you declare a function has a returntype of numeric, 

then try to return a string, ColdFusion would display an error. So when you are designing the 

functions within CFCs, you must consider what type of data will be coming in and going out. 

 

Let's look a simple, but complete function. This function takes a single argument, SizeInBytes, 

and converts this number to the appropriate value in KB, MB, or GB, then returns the value as 

a string. 

<cfcomponent> 

 <cffunction name="FormatFileSize" returntype="string"> 

  <cfargument name="SizeInBytes" type="numeric"> 

 

  <cfset kbsize = Arguments.SizeInBytes/1024> 

  <cfif kbsize GTE 1024> 

   <cfset mbsize = kbsize/1024> 

    

   <cfif mbsize GTE 1024> 

    <cfset gbsize = mbsize/1024> 

    <cfset finalsize = "#DecimalFormat(gbsize)# GB"> 

   <cfelse> 

    <cfset finalsize = "#DecimalFormat(mbsize)# MB"> 

   </cfif> 

  <cfelse> 

   <cfset finalsize = "#DecimalFormat(kbsize)# KB"> 

  </cfif> 

  <cfreturn finalsize> 

 </cffunction> 

</cfcomponent> 

Component Self-Documentation 

CFC's produce their own self-generating documentation when viewed in a browser. To do this, 

your ColdFusion server must have Remote Development Services (RDS) enabled. Using just 

the name and returntype attributes, however, this documentation won't be very useful. You can 

use additional attributes of the <cfcomponent>, <cffunction>, and <cfargument> tags to 

provide more thorough and useful descriptions about the use and purposes of each functions: 

 

<cfcomponent> 

Displayname An alternative name for the component; the best practice is 

to have this be the same as the filename, including the 

same casing 

Hint A text description of what the CFC does 

 

<cffunction> 

Displayname An alternative name for the method 

Hint A text description of what the method does 

Access Determines how the method can be accessed: 

 remote - Allows the method to be used as a web 

service; for Flash interoperability, you must use the 

remote method 

 public - Allows the method to be used by any 

internal ColdFusion page; this is the default value. 

 private - Allows the method to be used only by other 

methods within the same component 

 package - Allows the method to be used only by 

other methods in the same component or other 

components in the same directory 

 
<cfargument> 

Displayname An alternative name for the argument 

Hint A text description of what the argument should be and how 

its purpose 

 

You can also use Dreamweaver to view the documentation of CFCs using the Components tab 

of the Application panel. 

 
Figure 1: Components Tab in Dreamweaver 

You can right click on the individual components, methods, and arguments to see additional 

details based on the attributes noted above. 

Invoking a Static Component using createObject() 

There are several ways you can call a component. Two we discussed in the introductory 

sessions are using the <cfinvoke> and <cfobject> tags. These are the most commonly used 

methods, particularly in simpler applications. For this session, however, we will use the 

CreateObject() method of invoking components as it invoked from within <cfscript> and it can 

be used to both instantiate a component and execute a method within a single line of code.  

 

CreateObject() is actually the simplest way of calling a component, however in some 

situations, the CreateObject() option may be unavailable for security reasons, particularly when 

dealing with third-party hosting. A well-designed application will probably used several of the 

available invoking methods. Beyond that, for the most part the use of CreateObject() versus 

<cfinvoke> is a matter of personal preference, though CreateObject() seems to be the most 

popular among developers. I'm inclined to follow Raymond Camden's method: 

 Use CreateObject() for instantiation and if the call is going to be straight forward 

 Use <cfinvoke> when the included arguments are determined by conditional logic, when 

method calls are dynamically determined, or when a CFC is primarily intended to be a 

collection of UDFs that do not require instantiation 

 

Now, let's look at a called Math.cfc which has two fairly self-explanatory methods: 

<cfcomponent> 

<cffunction name="TimesTen"> 

<cfargument name="valueIn" required="yes"> 

<cfreturn valueIn * 10> 

</cffunction> 

<cffunction name="AddNums"> 

<cfargument name="num1" required="yes"> 

<cfargument name="num2" required="yes"> 

<cfreturn num1+num2> 

</cffunction> 

</cfcomponent> 

The CreateObject() function takes two required parameters, a string which designates the type 

of object being instantiated (in this case "component") and the path to the file of the component 

using dot notation. So, if we wanted to call the TimesTen() method in our Math.cfc component, 

and presuming it was in the same directory, we would instantiating the component and invoke 

the method using 

<cfset answer = createObject("component","math").TimesTen(10)> 

To display the result returned from the method, we would call it just like any other variable: 

<cfoutput>#variables.answer#</cfoutput> 

You can invoke the AddNums() method the same way. If you need to invoke either of these 

methods multiple times, you can instantiate the component just once, then invoke the methods 

as needed. 

<cfset cfcMath = createObject("component","math")> 

<cfset answer1 = cfcMath.TimesTen(10)> 

<cfset answer2 = cfcMath.addNums(1,2)> 

Keep in mind that each time you instantiate a component, there is a performance penalty, so 

you should use the CreateObject() as few times as possible and instead storing the CFC 

instance using either a local variable (as above) or by caching it in a persistent variable scope 

such as Application, Session, or Server. We'll look at caching more later. 

 

If a component is not in the same directory as the calling page, you must specify its location 

relative to a ColdFusion mapping. For example, if Math.cfc is located at 

C:\coldfusion8\wwwroot\CF8advanced\components\math.cfc, we would use dot notation to tell 

the system where to find it: 

<cfset cfcMath = createObject("component", 

"CF8advanced.components.math")> 

A directory containing components is called a "package." The full path name, 

"CF8advanced.components.math", is the package name of the component. CFCs can be 

stored anywhere on the server as long as there is a valid ColdFusion mapping to it. It is a good 

practice to group all of your components in a single directory per application. Unless you need 

to invoke your CFC methods using AJAX or a web service, this directory can also be outside of 

your application's public web directory, which is useful for security purposes. 

 

Using the createObject() instantiation method, there are two ways to pass in arguments to the 

method being called: 

1. Using positional notation (arg1,arg2...argN), which must match the defined order of the 

<cfargument> tags within the method 

2. Using name-value pairs (var1=arg1,var2=arg2,...,varN=argN) 

 

<cfset cfcMath = createObject("component","math.cfc")> 

<!--- method 1: use positional notation ---> 

<cfset result1 = cfcMath.addNums(2,2)> 

<!--- method 2: use name/value pairs ---> 

<cfset result2 = cfcMath.addNums(num1=2,num2=2)> 

Part 2: Application.cfc and Application Event Handling 
As noted in our intro sessions, the old method of initializing global variables and running code 

on every page request that used to be done in Application.cfc has been replaced with the use 

of the component Application.cfc. Remember, always upper-case the A. With Application.cfm, 

the code within the file was dynamically executed at the top of each page request, while a 

second special file, OnRequestEnd.cfm, could be used to execute code at the bottom of a 

page request.  

 

Application.cfm was traditionally used to: 

 include application initialization code and set application settings 

 set global variables 

 initialize user sessions 

 log system events 

 handle application errors more gracefully 

 perform basic security validations 

As neither Application.cfm nor OnRequestEnd.cfm files had any formalized structure or 

methodology, their use and content varied widely across applications, making them more 

difficult to read and maintain. Application.cfm was also fairly limited in its abilities. 

Application.cfc, introduced with ColdFusion 7, has a more formalized structure than 

Application.cfm, and is a more powerful system as it allows for functions that execute at a far 

greater range of pre-defined events: 

1. Application Startup (onApplicationStart()) 

2. Application Shutdown (onApplicationEnd()) 

3. Session startup (onSessionStart()) 

4. Session timeout (onSessionEnd()) 

5. Before the start of each page request (onRequestStart()) 

6. At the end of each page request (onRequestEnd()) 

7. At the start of each page request (onRequest()) 

8. When an error occurs (onError()) 

9. When a CFM file is not found (onMissingTemplate()) 

As with the Application.cfm file, when a page request begins, ColdFusion will look for the 

nearest Application.cfc file, starting in the current directory, then scanning recursively 

backwards through the directory tree until it encounters the drive root. 

 
Figure 2: Application.cfc processing 

Application settings 

Just as before, you can set a number of variables within an Application.cfc to set up the basic 

environment of your application. These are declared underneath the opening <cfcomponent> 

tag and before the declaration of the first function using the THIS scope. 

 



Attribute Default Description 

Name none The application name, it can be any text string that meets ColdFusion's normal variable naming rules. The Name 

attribute acts as an indirect scope for all of your application, session, and client variables, so it must be unique for each 

application on a server to avoid conflicts. If using transparent sharing of application variables with JSP, you must use 

the unnamed scope (blank), but only one application on any server should use this. 

ApplicationTimeout System defaults Sets a specified period of inactivity after which the onApplicationEnd() method will 

be triggered and all application scope variables will be deleted from memory. 

ClientManagement System defaults Activates or deactivates support for client variables. 

ClientStorage System defaults Where client variables will be stored, either registry (windows), a datasource (preferred) or in browser-based cookies 

LoginStorage Cookie Where information generated by the 

<cflogin> / <cfloginuser> tags is stored; default is cookie but can be changed to session 

SessionManagement False Enable/disable support for session variables; off by default 

SessionTimeout System defaults Specified period of inactivity after which the onSessionEnd() method executes and all session variables for the current 

session are deleted 

SetClientCookies True Send CFID and CFTOKEN cookies to the browser for identifying session variables. 

SetDomainCookies False Set CFID/CFTOKEN cookies as domain-level cookies; should only be set to true if you need client variables to be 

shared across multiple servers under the same domain. 

ScriptProtect System defaults Determines whether ColdFusion should attempt to protect itself against cross-site scripting attacks. Available options 

are "none", "all", and a comma-delimited list of specific scopes to protect (such as "FORM, URL, CGI"). When Script 

Protect is enabled, the system will filter any incoming data from the specific scopes that match the regular expression 

contained in the {cf_root}/lib/neosecurity.xml file. By default, this will replace any incoming data that contains the HTML 

tags Object, Embed, Script, Applet, and Meta. This setting provides a minimum level of extra protection for your 

applications, but it should be considered a total solution for this type of exploit and should always be backed up by 

outputting content submitted by an untrusted user using ColdFusion HTMLEditFormat() function. 

CustomTagPaths System defaults If per-application settings are enabled in the ColdFusion Administrator, you can use this setting to enter a comma-

delimited string that determines the order in which specific directories will be searched when invoking a CFML file as a 

custom tag using the <cf_tagname> syntax. 

Mappings System defaults If per-application settings are enabled in the ColdFusion Administrator, this setting allows you to configured your 

application level ColdFusion mappings, using <cfset>: <cfset this.mappings["/stuff"] = "c:\stuff"> 



<cfcomponent> 

<cfset this.name="CafeTownsendStepBySteps"> 

<cfset this.sessionmanagement="True"> 

<cfset this.scriptprotect="All"> 

<cfset this.clientmanagement="False"> 

<!--- Application.cfc event handlers ---> 

</cfcomponent> 

Inheritance 

When you create components, you can use the EXTENDS attribute to specify its parent. If they 

are not in the same directory, the full path to the parent component must be specified using dot 

notation. 

<cfcomponent extends = "components.mycomponent"> 

Components can be extended as many times as you need, and child components can be 

parent components for others. This is typically used to make your local codebase an instance 

of a parent codebase. This can also be used to have Application.cfc be a child of another 

Application.cfc file - allowing for one global Application.cfc that is extended on a per application 

basis. One particularly useful application of this is in the realm of CMS developments, where 

different instances (websites) of the CMS can be made by extending the core Application.cfc. 

 

The child compent can reference methods in the parent component using the SUPER scope. 

In this example, onApplicationStart() in the child Application.cfc references the like-named 

method in its parent. 

Parent component 

<cfcomponent> 

<cffunction name="onApplicationStart"> 

<!--- CMS-specific business logic here---> 

<cfreturn true> 

</cffunction> 

</cfcomponent> 

Child Component 

<cfcomponent extends="CF8advanced.cms.Application"> 

<cffunction name="onApplicationStart"> 

<!--- application-specific code ---> 

<cfset application.datasource = "CF8advStepByStep"> 

<!--- activate the cms codebase ---> 

<cfset super.onApplicationStart> 

<cfreturn true> 

</cffunction> 

</cfcomponent> 

Configuration with INIs 

You can use the Application.cfc to read an initialization file, such as an XML or INI file and set 

your application scope variables that will be set for the life of the application. Moving 

application-specific configuration information into an INI or XML file that is external to the 

application offer greater flexibility when deploying applications as compiled J2RR EAR/WAR 

files. These initialization files should site outside of your production application's web root for 

security. 

 

ColdFusion has made it easy to read configuration values from an INI file thanks to the 

GetProfileString() function. It takes three parameters: 

<cfset iniValue = GetProfileString(inipath, section, entry) /> 

1. inipath - the absolute path of the INI file; i.e. 

C:\CF8advanced\onYourOwn\Application.ini 

2. Section - the section of the initialization file where entry is located 

3. Entry - the name of the value to obtain 

 

Let's say we had a file named myapp.ini, located in the root of our C: drive that contains: 

[Global] 

Datasource=CF8advStepByStep 

BaseHref=/CF8advanced/stepByStep/ 

To initialization the variable application.datasource to the corresponding value in the INI file, 

we would use:  

<cfset application.datasource = GetProfileString("c:\myapp.ini", 

"Global", "Datasource"))> 

Application Event Handling 

Using the Application.cfc component, you can respond to two application specific events: 

onApplicationStart and onApplicationEnd. Code within the onApplicationStart() function will 

automatically be processed after a system restart when the first page of your application is 

requested, the first CFC is invoked, or the first request for a web service is made. This is also 

true after an Application time out. 

 

Typically, this function contains code that initializes global variables, do any logging desired, 

and deny access to the application if critical resources are down at the time of the restart (such 

as lack of database connectivity). This function must always return a Boolean value, either true 

if should continue processing or false to halt execution of the request. 

 

In this example, we get the datasource name from an INI file and cache it as an application 

variable for use throughout the application, and log if we couldn't find it. 

<cffunction name="onApplicationStart" returntype="Boolean"> 

<cfset var inipath = findfile("application.ini")> 

<cfif not IsDefined("application.datasource")> 

<cflog file="#this.name#" type="information" 

text="Startup"> 

</cfif> 

<cfset application.datasource = trim(GetProfileString(inipath, 

"Global", "datasource"))> 

<cfreturn true> 

</cffunction> 

 

Code contained within the onApplicationEnd() function is executed with the ColdFusion server 

is shutting down (gracefully) or the Application times out. It is primarily used for event 

logging/notification and doing delayed writes of cached information to a database. Unlike 

onApplicationStart(), it does not return any value. By its nature, it can not be used to return any 

interactive data to a web page and any Application variables you need to access must be 

using the passed-in argument AppScope. In this example, we are logging that the application 

shut down. 

<cffunction name="onApplicationEnd" returntype="void"> 

<cfargument name="AppScope" required="True" /> 

<cflog file="#this.name#" type="information" text="Shutdown" /> 

</cffunction> 

Logging with <cflog> 

As you might have noticed in those last examples, ColdFusion offers a tag called <cflog> 

which allows you to easily write messages to log files that can be managed through the 

ColdFusion Administrator. In the physical log, the entries are stored in a comma-separated 

value format in this order: type, thread, date, time, application, text. With the <cflog> tag, the 

following attributes are available, but only the text attribute is required. 

Attribute Description 

Text The text to be written to the log file 

Log Determines which ColdFusion log the entry will be written to; possible values are 

Application or Scheduler 

File Supersedes the log attribute and allows you specify a specific file, excluding the 

extension, to write to; this file will be located within the default log directory set in the 

ColdFusion Administrator 

Type The severity of the message which can be used for filtering when viewing the log; 

possible values are: Information, Warning, Error, and Fatal 

Application If yes, includes the name of the application from the Application.cfc file in the log; if 

set to no, it does not. 

 

Request Events 

Whenever a page is requested from the ColdFusion server, it will process any code within 

Application.cfc's onRequestStart() function first. This is essentially the same behavior that 

Application.cfm used to perform. Typically, this function is used to contain code that deals with 

evaluating access rights to a page or subdirectory, redirect users to a login page is needed, 



SDG Advanced ColdFusion Training, Session 1 

Page 17 of 27 

and performs other redirection/method invocations based on available URL and/or Form 

variables. As with onApplicationStart(), it must return a Boolean value of true to continue 

processing, or false to halt. The function is automatically passed an argument by the system, 

TargetPage, that contains the name of the requested page from the web root. 

 

In this example of OnRequestStart, we look for a URL parameter named init. If its found, it will 

restart the application without having to restart your ColdFusion server (a very useful function 

indeed!).  

<cffunction name="onRequestStart" returntype="Boolean"> 

<cfargument type="String" name="TargetPage" required="true" /> 

<cfif IsDefined("url.init")> 

<cfset onApplicationStart()> 

</cfif> 

<cfreturn true> 

</cffunction> 

Another request event handler is onRequest(), which contains code to be processed at the 

start of the page request, immediately after onRequestStart(). Typical uses include capturing 

the contents of a page and doing post processing, such as removing white space, or using 

business logic to redirect users to different pages. It does not return any value and is 

automatically passed a single argument by the server, TargetPage.  

 

When using onRequest(), keep in mind that you must explicitly execute the target page, 

usually through a cfinclude, as the function will basically execute in place of the original page 

request. Using OnRequest will prevent any web services, event gateways, and Flash remoting 

requests that fall within its scope to fail. 

 

Here, we use onRequest to strip all the extraneous white space from pages before they sent to 

the client browser. 

<cffunction name="onRequest" returntype="Boolean"> 

<cfargument type="String" name="TargetPage" required="true" /> 

<cfsavecontent variable="pagecontent"> 

<cfinclude template="#arguments.targetpage#"> 

</cfsavecontent> 

<cfoutput>#reReplace( pagecontent, "[[:space:]]{2,}", chr( 13 ), 

"all" )#</cfoutput> 

<cfreturn true> 

</cffunction> 

 

At the end of a page request, the server will automatically process any code within your 

Application.cfc's onRequestEnd() function. This function replaces the old-style 

OnRequestEnd.cfm construct used in earlier versions of ColdFusion, and if you are using 

Application.cfc, any OnRequestEnd.cfm templates will be ignored. The most typical use of this 

function is to include a common footer on all of your application pages. You do, however, need 

to take care to wrap any generated output in the appropriate conditional blocks to ensure it 

only runs on regular pages, rather than on pages that output XML or return results from a web 

service invocation. It can also be used for fulfilling auditing requirements and generating 

statistics on page rendering times. As with onRequest(), it does not return a value and is 

passed the argument TargetPage. 

<cffunction name="onRequestEnd" returntype="void"> 

<cfargument type="String" name="TargetPage" required="true" /> 

<cfif ListLast(arguments.targetpage,".") is "cfm"> 

<cfoutput> 

Copyright &copy; #year(now())#. 

Cafe Townsend Inc. 

</cfoutput> 

</cfif> 

</cffunction> 

Part 3: Locking Shared Scopes 
Much of the processing that Application.cfc does involves manipulating server, application, and 

session variables. Once you create any of these variables, they are stored in the server's 

memory and are available across multiple pages. As the ColdFusion server is multithreaded 

and is usually processing multiple pages at the same time, multiple requests may want to 

access these variables at the same time. Prior to ColdFusion MX, you had to manually lock 

your various shared-scope variables every time they were read from or written to to protect 

them from conflicts. Leaving them unlocked could result in memory corruption, server 

instability, and system crashed. Fortunately, ColdFusion now automatically makes these data 

structures thread-safe and will make multiple pages requesting access "wait in line" before 

their turn. This avoids corruption issues, but does not necessarily mean you will get the results 

you are desiring, so while locking is no longer necessary to the extent it used to be, it does still 

have its place in ColdFusion applications. 

 

Let's say you had an application variable named application.sessioninfo.logincount that you 

are using to see store the current number of logged in users on your site. Within your 

Application.cfc, you initialized it using  

<cfset application.sessioninfo = structNew()> 

<!--- other initialization code ---> 

<cfset application.sessioninfo.logincount = 0> 

In a reporting page, you have the following code to view this value: 

<cfoutput> #application.sessioninfo.logincount# </cfoutput> 

If we have two concurrent requests, user A triggers the reinitialization of 

application.sessioninfo as an empty structure while user B is simultaneously attempting to 

output the contents of application.sessioninfo.logincount, user B would receive a variable not 

found error. You can control this type of simultaneous access using the <cflock> tag. This tag 

gives ColdFusion a signal that each page request should have exclusive access to its code for 

the duration of the tag. Of course, this lock only works  if each and every page referencing the 

shared data also locks it, whether it is reading the data or writing it. 

<cflock scope = "Application/Server/Session" 

type = "readOnly/Exclusive" 

timeout = "timeout in seconds" 

ThrowOnTimeout = "Yes/No" 

name="lockname"> 

[ statements using shared-scope variable ] 

</cflock> 

Using <cflock> you can perform two kinds of locks: an exclusive lock, for writing shared data, 

and a read-only lock when you just need to read the shared data.  In all, the <cflock> tag has 

five attributes available. 

Attributes Description 
Scope The shared scope to lock; it is mutually exclusive with the name 

attribute 

Type Read-only or exclusive 

Timeout Maximum number of seconds the server should wait to obtain the 

lock; if its obtained, the system continues, otherwise it acts 

according to the ThrowOnTimeout attribute 

ThrowOnTimeout Yes if the system should throw an error if it can not lock the scope, 

otherwise no and tag block is skipped 

Name Lock name that can be used instead of specifying a scope. Use of a 

name only locks access to the specific shared score variables inside 

the tag - instead of locking access to the entire scope. Lock names 

are global to a ColdFusion server and cannot be equal to an empty 

string. 

 

Exclusive locks 

With exclusive locks, single-threads can access the shared data within the <cflock> tag - no 

other page requests will be able to read or write any of the locked variables until they are 

unlocked, which is determined by how long it takes the code with the tags to execute. It only 

works when ever reference to the data is locked, and exclusive locks are issued on a first-

come, first-served basis. 

 

Revisiting our previous example, let's take care of that situation of conflict using a <cflock> 

<cflock scope="application" type="exclusive" timeout="10"> 

<cfset application.logincount++> 

</cflock> 

Read-Only locks 

A read-only lock will allow multiple requests to access shared data inside the <cflock> tag, but 

only for the purposes of reading it. It will not allow the shared data to be modified. If any other 

request already has an exclusive lock on the shared data, the read-only request will still have 

to wait until the exclusive lock is released. In our example for outputting the 

application.logincount variable, we would use a read-only lock to just get the most up-to-date 

value: 

<cflock scope="application" type="readOnly" timeout="10"> 

<cfoutput> 

Number of logins: #application.logincount# 

</cfoutput> 

</cflock> 

If you do not use a lock, you may not always get the most recent value because someone may 

be updating at the time call. While in some cases, this may not matter, you should always 

consider whether this is of importance in your specific situation. 

Named Locks 

Exclusively locking every application variables will cause performance to suffer. Every time a 

user changes the variable, no other locked references can be executed. To reduce this 

problem, you can use named locks instead of scoped locks. Named locks only lock the specific 

shared-scope variables inside the tag. Let's say we always want to count the number of users 

who explicitly log out of your site versus those who log in. On the login page, you have: 

<cflock name="cafeTownsend.Logincount" type="exclusive" timeout="10"> 

<cfset application.logincount++> 

</cflock> 

On the logout page, you have: 

<cflock name="cafeTownsend.Logincount" type="exclusive" timeout="10"> 

<cfset application.logincount--> 

</cflock> 

Now, keep in mind, that the lock names are global across the server, so its always best to 

include the application name in the lock name to avoid conflicts across applications. 

On your display page, you have: 

<cflock name="cafeTownsend.Logincount" type="readOnly" timeout="10"> 

<cfoutput> 

Number of logins: #application.logincount# 

</cfoutput> 

</cflock> 

Of course, named locks are harder to keep track of than scoped locks, and you'll have to 

remember to use them for all uses of the variable. Carefully weigh the performance benefits 

you can gain by using named locks versus the errors you might introduce by having to 

remember and use the correct names. 

 

When deciding when to use locks, consider the variable's usage: 

 If its written once, and read many times, such as application.datasource, locks are 

generally not needed 

 Variables written many times and read many times, should be examined on a case-by-

case basis to determine if locking is needed to maintain integrity 

 

On the whole, you should try to minimize the use of locks. Only lock data that change regularly 

and whose value or meaning would be corrupted if the processing were not single-threaded. 

Avoid using time-consuming processes within the <cflock> tag, do only what you need to do 

and get back out quickly. Use named locks for finer control of locks, but make sure to include 

the application's name in its lock name. And never, ever nest locks. 

Part 4: Session Tracking and Session Event Handling 
As with application and requests, ColdFusion offers functions within Application.cfc for running 

on the start and end of each user session. onSessionStart() is usually used to initialized 

complex session data structures or creating session-based CFC instances. You do not need to 

lock any of the session variables instantiated by this function as it is guaranteed to be single-

threaded by the server. The function does not return any value. 

 

The following example initializes an application variable that tracks the number of active 

sessions. It also logs the start time for the session. 

<cffunction name="onSessionStart" returntype="void"> 

<cflog text="Session Start:#session.sessionid#" 

type="information" file="#this.name#"> 

<cflock name="cafetownsendoyourown_activesessions" 

type="exclusive" timeout="30"> 

<cfset application.activeSessions++> 

</cflock> 

</cffunction> 

The code within the onSessionEnd() function runs whenever a user session expires due to 

inactivity. It is generally used for logging purposes or writing session information to the 

database. It does not run when the application times out nor during a server shutdown as it is 

user dependant. It is passed two arguments, SessionScope containing the user's session 

scope variables, and AppScope, containing the application variables. 

 

Taking the example above and building on it, this example reduces the number of active 

session on the application time out and logs its ending. 

<cffunction name="onSessionEnd" returntype="void"> 

<cfargument name="SessionScope" type="Struct" required="True" /> 

<cfargument name="AppScope" type="Struct" required="False" /> 

<cflog file="#this.name#" text="Session 

End:#arguments.sessionscope.sessionid#" type="information"> 

<cflock name="cafetownsendonyourown_activesessions" timeout="30" 

type="exclusive"> 

<cfset arguments.AppScope.activeSessions--> 

</cflock> 

</cffunction> 

Best Practices and Reminders 
1. Component properties must be declared immediately after the opening <cfcomponent> 

tag. 

2. When developing functions, carefully plan for the type of data it will accept and will 
return, as ColdFusion strictly enforces these data types. 

3. Function specific variables should be declared using the var attribute of the <cfset> tag. 

4. Make use the self-documenting features of CFCs for both your own documentation and 
when sharing with others. 

5. When setting the DisplayName of components, use the same name as the CFC file, 
including the same casing. 

6. When using CreateObject(), invoke a component only as needed and instead 
cache/save an instantiation for reuse whenever possible 

7. Components can inherit properties and methods from other components using the 
EXTENDS attribute. 

8. Always use Application.cfc, with a capital A, over Application.cfm and 
OnRequestEnd.cfm 

9. Application.cfc uses an event-driven model to handle requests. 

a. onApplicationStart() is triggered when your application is accessed for the first 
time following a restart. 

b. onApplicationEnd() runs when the application times out or during a system 
shutdown 

c. onRequestStart() runs when a page is requested. 

d. onRequest() runs after the onRequestStart() method at the start of a page 
request. It must not be defined for CFC's that implement web services, event 
gateway, or Flash Remoting requests. 

e. onSessionStart() is typically used to initialize session variables for a specific 
browser instance 

f. onSessionEnd() executes when the user's session times out or, if using J2EE 
sessions, when the user closes their browser, but it does not execute when the 
application or server shuts down 

10. When using INI files, store them outside of the production web folders 

11. Use ScriptProtect for low level cross-script protection, but never rely on it to be your 
application's only protection 

12. Use <cflog> to create custom logs for your individual applications, allowing for more 
robust and usable information entries 

13. When using onRequest(), remember to explicitly call the target page, usually using 
<cfinclude>; also remember it cannot be used with Flash remoting, event gateways, and 
web services 

14. Though <cflock> is no longer required for most applications, consider its use in cases 
where variables are written and read frequently and integrity of the value is critical 

15. Keep the code within the <cflock> simple and basic, don't include unrelated or time-
consuming processes like queries 

16. When using named locks, include the application name in its lock name to avoid cross-
application conflicts 

17. Never nest locks 

Applying What We Learned 
1. Create an application.cfc file 

a. Read in an .INI file that you create to set the application name and datasource 
name; make sure to use conditional logic to check for the existence of the setting 
first 

b. Enable session management and set the session timeout to 20 minutes 

c. Disable client variables within your application 

d. Set the LoginStorage attribute to session 

e. Turn on domain cookies attribute 

f. Turn on cross-site script protection for all variable scopes 

g. Set the inactivity timeout for your application variables to 1 hour. 

h. Include the on start/end application, request, and session events 

i. Log the application's starting and stopping 

ii. Use the request event handlers to include common headers/footers in 
your application 

iii. Use the session event handlers to track the number of active sessions on 
your application 

2. Create a component with three functions 

a. First, format a passed in number as a percent 

b. Second, retrieve records from the sdg_training database's ml_series table, with 
an optional limit to a single record 

 

Figure 3: Schema of ml_series 
c. Third, pull all the authors for a requested titleid and format them in a comma 

separated list with firstname and last name; include "and" before the last author 

Figure 4: Schema of ml_bookstoauthors 

 
Figure 5: Schema of ml_authors 

3. Invoke each function of your component in a cfm page using either <cfinvoke> or 
CreateObject(), as is appropriate 


